

HT45R36 使用普通 I/O 驱动蜂鸣器


文件编码：HA0119S

简介

有些 MCU 由于没有 PFD 功能，但在实际应用中需要蜂鸣器发声，所以本范例以 HT45R36 为例，用普通 IO 仿真 PFD 功能，驱动 2K 的蜂鸣器。

应用电路图

将蜂鸣器直接接于 IO（PA3）和地之间。通过控制 IO 的输入输出来控制 BZ 的输出。



应用电路图

软件流程说明

```

.listinclude
INCLUDE HT45R36.INC
;;*****
DATA_MAIN .SECTION 'DATA'           ;定义数据缓存器
    R_ACC_BAK      DB ?              ;累加器数据备份
    R_STATUS_BAK   DB ?              ;状态缓存器数据备份
    R_COUNTER1     DB ?              ;定时计数器 1
    R_COUNTER2     DB ?              ;定时计数器 2
    F_HALFSECOND   DBIT             ;0.5s 状态旗标
;-----
CODE_MAIN .SECTION AT 00H 'CODE'     ;程序段
    JMP MLP_MAIN
    ORG 04H          ;中断 0 入口地址
    RETI
    ORG 08H          ;中断 1 入口地址
    RETI
    ORG 0CH          ;定时/计数中断入口地址
    JMP ISR_TIMER   ;跳到定时/计数中断程序
    ORG 10H          ;RC 振荡中断入口地址
    RETI

ISR_TIMER:                ;定时/计数中断程序
    MOV R_ACC_BAK, A      ;备份累加器资料
    MOV A, STATUS
    MOV R_STATUS_BAK, A  ;备份状态缓存器数据
    MOV A, 08H
    XORM A, PA            ;PA3 取反输出
    INC R_COUNTER1       ;进中断一次定时计数器 1 加一
    MOV A, R_COUNTER1
    XOR A, 80
    SNZ Z                ;判断是否计数到 (80*125us) =10ms
    JMP L_EXIT_ISR_TIMER ;没有计数到 10ms, 退出中断
    CLR R_COUNTER1       ;已计数到, 清除定时计数器 1 以便下次计数
    INC R_COUNTER2       ;每计数到 10ms 时计数器 2 加一
    MOV A, R_COUNTER2
    XOR A, 50
    SNZ Z                ;判断计数器 2 是否计数到 (50*10ms) =0.5s
    JMP L_EXIT_ISR_TIMER ;没有计数到 0.5s, 退出中断
    CLR R_COUNTER1       ;已计数到, 清除定时计数器 1 以便下次计数
    CLR R_COUNTER2       ;已计数到, 清除定时计数器 2 以便下次计数
    SZ F_HALFSECOND      ;判断 0.5s 状态旗标, 并取反
    JMP L_CLRFLAG        ;0.5s 状态旗标已置起, 跳到清 0.5s 状态旗标
    SET F_HALFSECOND     ;置起 0.5s 状态旗标
    JMP L_EXIT_ISR_TIMER ;退出中断

L_CLRFLAG:
    CLR F_HALFSECOND     ;清除 0.5s 状态旗标

L_EXIT_ISR_TIMER:
    MOV A, R_STATUS_BAK  ;还原状态缓存器数据
    MOV STATUS, A
    MOV A, R_ACC_BAK     ;还原累加器数据
    
```

```

RETI
MLP_MAIN:
MOV    A, 09H                ;
MOV    INTC0, A              ;打开定时/计数器中断
CLR    F_HALFSECOND         ;初始化 0.5s 状态旗标,
CLR    R_COUNTER1           ;初始化定时计数器 1
CLR    R_COUNTER2           ;初始化定时计数器 2
SET    PAC.3                 ;设置 PA3 为输入, 关闭 PFD
MOV    A, 06H
MOV    TMR, A                ;设置计数初始为 6, 计数 250 次进一次中断
MOV    A, 91H
MOV    TMRC, A               ;设置为内部定时, 分频 2MHz/2=1MHz,
                                ;250us 进一次中断
LOOP:
SZ     F_HALFSECOND         ;判断 0.5s 状态旗标是置起
JMP    L_PFDOPEN            ;是, 跳到打开 PFD
L_PFDCLOSE:
SET    PAC.3                 ;否, 设置 PA3 为输入关闭 PFD
JMP    LOOP                  ;循环判断是否输出 PFD
L_PFDOPEN:
CLR    PAC.3                 ;是, 设置 PA3 为输出打开 PFD
JMP    LOOP                  ;循环判断是否输出 PFD
    
```

软件说明

这段程序的功能是蜂鸣器每隔 0.5s 叫一次，叫声持续 0.5s。

程序通过设置定时 250us 进一次中断，当 PA3 为输出时，PA3 进中断产生 2K 的方波以便驱动蜂鸣器发声。

当设置 PA3 为输入时，则 PA3 无法驱动蜂鸣器，蜂鸣器停止发声。

程序中通过定时计数器 R_COUNTER1, R_COUNTER2 来定时时间 0.5s, 0.5s 到就取反 0.5s 状态旗标，通过其旗标来循环开关蜂鸣器。